

## **Backsplash Patterns for Your World: A Look at SAS® OpenStreetMap (OSM) Tile Servers**

Barbara B. Okerson, Manteo, NC

### **ABSTRACT**

Originally limited to SAS Visual Analytics, SAS now provides the ability to create background maps with street and other detail information in SAS/GRAPH® using open source map data from OpenStreetMap (OSM). OSM provides this information using background tile sets available from various tile servers, many available at no cost. This paper provides a step-by-step guide for using the SAS OSM Annotate Generator (the SAS tool that allows use of OSM data in SAS). Examples include the default OpenStreetMap tile server for streets and landmarks, as well as how to use other free tile sets that provide backgrounds ranging from terrain mapping to bicycle path mapping. Dare County, North Carolina is used as the base geographic area for this presentation.

### **INTRODUCTION**

OpenStreetMap is a source of mapping data built by a community of mappers that contribute and maintain data about roads, trails, cafés, railway stations, and much more, all over the world. OSM is open data and can freely be used for any purpose as long as OpenStreetMap and its contributors are credited. In other words, OSM is a collaborative project to create a free and editable map of the world. It is supported by the OpenStreetMap Foundation, a non-profit registered in England and Wales.

The OSM project was started because geographic data is not free in many parts of the world. For example, Google maps data is copyrighted by many organizations and licensed by Google, therefore not free to use without permission. Created by Steve Coast in the UK in 2004, OSM was inspired by the success of Wikipedia and the predominance of proprietary map data in the UK and elsewhere. OSM currently has over two million registered users who collect data using manual survey, GPS devices, aerial photography, voice recorders and other free sources.

SAS provides a tool called the OSM Annotate Generator to enable SAS users to use OSM data in SAS. While the OSM Annotate Generator is written in SAS/AF® SCL code, it does not require a SAS/AF license to run. The OSM Annotate Generator creates an annotate data set containing a static background map composed of tiles or square bitmap graphics displayed in a grid arrangement to show a map. Map tiles are typically 256x256 pixel images.

### **SAS OSM ANNOTATE GENERATOR**

The SAS OSM Annotate Generator creates an annotate data set containing a static background map that can be used with PROC GMAP, PROC SGMAP and other SAS mapping tools. This tool retrieves the required background map images from the selected tile server(s) and stores them on your local disk while creating the annotate data sets that use the images.

The area displayed on the map is specified either through: 1) a call to or subset of a map in SAS map data sets (either traditional or GfK), or 2) through inclusion of latitude/longitude information. The resulting background OSM maps can display data in addition to the OSM information. In both situations, SAS uses the data sets that were created as the map background.

SAS provides the default OSM background tiles, but it is possible to use a different set of background map tiles other than the set provided by SAS. In order to do this, you must specify a different map tile server. Before using any map tile server, be sure to read the Terms of Use documentation to make sure you have a legal right to use the tiles. Each tile server has its own copyright message; you need to apply the proper copyright message to your map. This is required by the owner for use of the maps.

As mentioned above, the OSM Annotate Generator is a SAS/AF SCL application. Values are passed to the Annotate Generator by setting special macro variables with a series of %lets. Here are the macro variables accepted by the application:

- **\_inmapds** – Input map data set (polygonal data)  
OR **\_inannods** – Input annotate data set.
- **\_imageloc** – Location where the map tile set images will be downloaded and saved.
- **\_outannoOSMds** – Output annotate data set containing the background map.
- **\_xpixels** – xpixel size of the background image. Defaults to 800.
- **\_ypixels** – ypixel size of the background image. Defaults to 600.
- **\_autoproject** – Turn on the projection to Mercator to match what is needed for OSM.
- **\_outmapds** – Output map data set
- **\_outannods** – Output annotate data set.
- **\_backgroundonly** – Turn on the background only option.
- **\_tileserv** – Location of the map tile server being used if other than the default server.
- **\_copyright** – Set the text of the copyright string for alternate **\_tileserv** settings.
- **\_proxy** – Set your proxy server if needed.

To run the OSM Annotate Generator, specify the following:

```
PROC DISPLAY cat=OAG.osmmapanno.osm_anno_gen.scl;  
Run;
```

Values set with the macro variables stay set until they are changed or cleared. If you forget to clear them, you may get unexpected results or errors. SAS provides a program called `OAG_reset.sas` that should be run before each program. The OSM Annotate generator and supporting programs are available at: <http://support.sas.com/rnd/papers/sasgf13/GooglelikeMaps.zip>.

## EXAMPLES CREATING MAPS WITH SAS AND OSM TILE SERVERS

The examples that follow use different tile servers to create maps in SAS. Dare County, North Carolina is the selected area for these maps. Dare County is the easternmost county in the state of North Carolina. The county is named after Virginia Dare, the first child born in the Americas to English parents, who was born in what is now Dare County. As of the 2010 census, the population of the county was 33,920; however the county's tourism industry results in a large seasonal population with an average daily population from June through August of between 225,000 to 300,000 people. The county seat is Manteo on Roanoke Island which is about 200 miles east of Raleigh.

Examples included in this paper use four tile servers as follows:

- SAS default tile server.
  - Default OSM tile server (roads and features).
  - OSM tile server with choropleth map.
- Stamen tile server.
  - Terrain.
  - Watercolor.
- CARTO tile server.
  - Black and White.
- Thunderforest tile server.
  - Bike Path.
  - Background.
  - Spinal.

### EXAMPLE ONE – DARE COUNTY AND ENVIRONS

The map below is created by subsetting the SAS GfK North Carolina map for Dare County and overlaying that information transparently over the OSM background. The GfK coordinates dictate the OSM area; however the background image will also include the nearby area.



**Figure 1. Dare County area with OSM.**

Here is the code used to create the map above. Note the use of the OSM Annotate Generator macro values. The pattern color A00000000 equates to completely transparent so all of the OSM features remain visible. This color pattern is part of the RGBA color codes available beginning with SAS 9.3. Color names are in the form of ARRGGBBAA, as below:

- A – RGBA color specification.
- RR – red component.
- GG – green component.
- BB – blue component.
- AA – transparency component.

```

/* File, Directory and Program Locations*/
filename odsout '/cma/nshare/bokerson/OSMMaps';
libname OAG '/cma/nshare/bokerson/OSMMaps/AFpgm';

/* reset the macro variables */
%include '/cma/nshare/bokerson/OSMMaps/AFpgm/OAG_reset.sas';

/* Subset map for Dare County*/
data daremap; set mapsgfk.us_counties (where=(statecode='NC' and county=055));
  x=long; y=lat;
run;

/* Set the macro variables for the OSM Generator */
%let _inmapds=daremap;
%let _outmapds=work.mdaremap;
%let _outannoOSMds=work.mannoosm;
%let _imageloc='/cma/nshare/bokerson/OSMMaps/';

```

```

%let _xpixels=2400;
%let _ypixels=1800;
%let _autoproject='Y';

/* Set xpixels, ypixels to the size to be used */
goptions xpixels=&_xpixels ypixels=&_ypixels;

/* Generate background image */
PROC DISPLAY cat=OAG.osmmapanno.osm_anno_gen.scl; run;

/* Generate the map */
GOPTIONS DEVICE=png;
ODS LISTING CLOSE;
ODS HTML path=odsout body="ChoroDCGfK_png..html";
pattern; /*reset pattern statements */
pattern v=me c=A00000000 r=100;

PROC GMAP map=&_outmapds anno=&_outannoOSMds data=daremap all;
  id county;
  choro statecode /nolegend name="ChoroDCGfK";
run; quit;
ODS HTML CLOSE;
ODS LISTING;

```

## EXAMPLE TWO – CHOROPLETH MAP WITH OSM ANNOTATE GENERATOR

One of the most useful applications of OSM maps and the OSM Annotate Generator is the ability to map data on a base map that includes highways and other features without using specialized GIS software. To create a choropleth map, a response data needs to be created and/or processed. This data set needs to be tied to a SAS/GRAPH map data set.




Figure 2. Dare County Choropleth Map with OSM.

Partial code is listed below, illustrating the pattern used, the legend statement and the GMAP procedure. It is important to use transparent colors when creating choropleth maps with the OSM map tiles so as not to lose the OSM information. Legends can be inside or outside of the map area. Using the county field as the ID variable allows distinguishing what is actually Dare County from the surrounding areas.

```

/* Set transparent colors */
pattern1 v=solid repeat=1 c=A00ff00aa;


/* Create a legend */
legend1 label=(position=top justify=center
font="albany amt" height=1 " ")
shape=bar(.1in,.1in)
value=(justify=left height=1 font="albany amt"
'Dare County, North Carolina');

/* Draw the map */
PROC GMAP map=&_outmapds anno=&_outannoOSMds data=mymap all;
id county;
choro statecode /legend=legend1 name="ChoroFlGfK";
run; quit;

```

### EXAMPLE THREE – STAMEN TILE SERVER TERRAIN

The map below is uses the same subsetting of Dare County, North Carolina as the previous map, but uses map tile set data from Stamen Design. Stamen is a San Francisco design and development studio focused on data visualization and map-making that provides free tile sets that can be used with OpenStreetMap data worldwide. The example map images below use the Terrain map tile set for Dare County.



**Figure 3. Dare County Area Terrain.**

The examples previous to this one used the default OSM tile server and therefore it was not necessary to set the tile server or copyright macro variables. Because more than one map tile server is available, use of these require identifying the server location and using the correct copyright message on your maps.

You also must make sure that you have a legal right to use the tile set that you have selected. The non-default servers may run slower in SAS because they are not already set up for SAS use. The two new macro variables needed for a non-default tile server are: `_tileserv` and `_copyright`.

To point the `_tileserv` macro variable to an alternate server, specify the following:

```
%let _tileserv= <tileserver>;
```

You must also specify the proper copyright message required by the owner of the map tile set. These messages are provided with the map tile sets. Use the `_copyright` macro variable as follows:

```
%let _copyright= <copyright-statement>
```

Here is partial code for this map. Note the values for `_tileserv` and `_copyright`. Be sure to include the final slash on the `_imageloc` macro variable as it is required. The background option on the `_tileserv` macro limits to background only, removing labels and streets. To include labels and streets, the statement would be as follows: `%let _tileserv=tile.stamen.com/terrain;`

The below code snippet highlights use of the SAS code for reset of macro variables. This is used in all of my OSM code. As mentioned previously, values set with the macro variables stay set until they are changed or cleared, so this reset should be part of every run.

```
/* reset the macro variables */
%include '/cma/nshare/bokerson/OSMMaps/AFpgm/OAG_reset.sas";

/* set the macro variables */
%let _inmapds=daremap;
%let _outmapds=work.mdaremap;
%let _outannoOSMds=work.mannoosm;
%let _imageloc='/cma/nshare/nophi/AA92483/SGF2017/GooglelikeMaps/';
%let _xpixels=2400;
%let _ypixels=1800;
%let _autopproject='Y';
%let _tileserv=tile.stamen.com/terrain-background;
%let _copyright="Map tiles by Stamen Design,Data by OpenStreetMap.";

/* Set xpixels, ypixels to the size to be used */
goptions xpixels=&_xpixels ypixels=&_ypixels;

/* Generate background image */
proc display cat=OAG.osmmapanno.osm_anno_gen.scl; run;

GOPTIONS DEVICE=png;
ODS HTML path=odsout body="DareTerrain_png..html";
pattern v=me c=A00000000 r=100;

/* Draw the map */
proc gmap map=&_outmapds anno=&_outannoOSMds data=daremap all;
    id county;
    choro statecode /nolegend name="DareTerrain";
run; quit;
ODS HTML CLOSE;
```

#### EXAMPLE FOUR – STAMEN TILE SERVER WATERCOLOR

In addition to the terrain map tile data, Stamen provides other map tile data sets. Those currently available worldwide are:

- Toner and Toner 'lite' – High contrast black and white.
- Watercolor - Raster effect area washes and organic edge.
- Burning – Animated burning effect. This requires a WebGL-enabled browser, such as Google Chrome.

Stamen also provides local San Francisco map tiles (trees, cabs, crime) and has a Mars surface map tile set available. The example below uses the Watercolor map tile data set for Dare County.



**Figure 4. Dare County Area Watercolor.**

One of the goals of the Watercolor map was to capture some of the irregularities of a handmade object. Starting with OpenStreetMap vector data, all styling is done in raster space. This allowed access to overlaps of objects without increasing processing time for denser areas. This also allowed for treating ground, water and green space differently. Information on the creation process can be found at: <https://hi.stamen.com/watercolor-process-3dd5135861fe>.

The only required change to the SAS code for this map from the map above is to the `_tileserv` macro variable as illustrated below:


```
%let _tileserv=tile.stamen.com/watercolor;
```

#### **EXAMPLE FIVE – CARTO FILE SERVER**

CARTO is a service cloud computing platform that provides Geographic Information System (GIS) and web mapping tools for display in a web browser. The company has positioned itself as a Location Intelligence (LI) platform due to tools for data analysis and visualization that do not require previous GIS or development experience. CARTO users can use the company's free platform or deploy their own instance of the open source software. CARTO is offered as a freemium service, where accounts are free up to a certain size.

Currently CARTO serves banking and finance, business intelligence/analytics, government, real estate, web development, journalism, natural resources, earth observation, non-profits, and education industries. CARTO was formerly CartoDB, Inc. and changed its name to CARTO in July 2016. CARTO was founded in 2012 and is currently based in Brooklyn, New York with an office in Madrid, Spain.

Two map tile data sets are available for unlimited free use by CARTO - CartoLight and CartoDark. Both are natural light, low zooms, no labels. The example below is with CartoLight.



**Figure 5. Dare County Area CARTO Black and White.**

**EXAMPLE SIX – BICYCLE PATHS MAP USING THUNDERFOREST TILE SET**

Thunderforest is a project of Gravitystorm, LTD, based in the UK to create and style maps using vector technology. Vector tiles contain the map information required for image tiles and are created from shapefiles. By pre-processing the data into vector tiles they are easily converted into raster image tiles so that the raster images do not need to be stored.


To use tile sets available from Thunderforest, an API key is needed. API Keys are tokens that you append to each request to identify that the request is related to your account. They are 32-character identifiers, made of letters and numbers. Use of the data and tiles are free, but you do need to create an account and receive an API key. Unfortunately, the macro variables created by SAS as part of the OSM Annotate Generator do not accept API keys. Therefore, another method is required to create maps using these map tile data sets.

To create a map using Thunderforest or other map tile data sets that require APIs, follow the steps below:

- Determine the minimum and maximum longitude and latitude values for the data.
- Determine the center of the area to be mapped.
- Define the zoom level (10 is maximum).
- Set the maximum x and y pixel counts.
- Convert the longitude and latitude degrees to x/y in the OSM tile-based coordinate system.
- Set the minimum and maximum values for the tiles.
- Download the map tiles using the API key.
- Combine the tile images into a SAS annotate data set.
- Flip the Y coordinate to start at top left rather than bottom left.
- Map the data with Proc GANNO.

The below map uses Thunderforest map tiles for bicycle paths in Dare County.





**Figure 6. Dare County Area Bicycle Routes with Thunderforest Map Tile Data.**

Partial SAS code is provided below. Because the OSM macro variables are not used, the footnote needs to be added to the annotate data set as a label statement. Also, note the link to the equations for tile-base coordinate systems.

```

/* determine the min and max values for the map data */
proc sql noprint;
  select min(x), min(y), max(x), max(y) into :minx, :miny, :maxx, :maxy
  from daremap;
quit;

/* determine the center of the map area */
data center;
  long=&minx+(&maxx-&minx)/2;
  lat=&miny+(&maxy-&miny)/2;
  output;
run;

/* Define a zoom level for the map tiles */
%let zoomlvl=10;
goptions xpixels=4800 ypixels=3600;

/*Convert the long & lat degrees to x/y in OSM tile-based coordinate system,
equations from: http://wiki.openstreetmap.org/wiki/Tilenames*/
data test;
  set center;
  lon_rad=(atan(1)/45)*long;
  lat_rad=(atan(1)/45)*lat;
  n=2**&zoomlvl;
  x = ((long + 180) / 360) * n;
  y = (1 - (log(tan(lat_rad) + (1/cos(lat_rad)))) / constant('pi')) / 2 * n;
run;

```

```

/*Set min and max values for the tiles */
data _null_;
  set test;
  call symput('min_x',int(x)-1);  call symput('max_x',int(x)+1);
  call symput('min_y',int(y)-1);  call symput('max_y',int(y)+1);
run;

/* Download each map tile - include api key*/
%macro download_map(zoom, x, y);
filename osm_pic url
"http://a.tile.thunderforest.com/cycle/&zoom./&x./&y..png?apikey=3c2ebc25cfac43
029c61cd8d0cad274b";
data _null_; n=-1;
  infile osm_pic recfm=s nbyte=n length=len _infile_=tmp; input;
  file "/cma/nshare/nophi/OSMMaps/daremap_&zoom._&x._&y..png" recfm=n;
  put tmp $varying32767. len;
run;
%mend;

/*Loop through the min and max values to download each tile */
data _null_;
do loop_x=&min_x to &max_x;
  do loop_y=&min_y to &max_y;
    call execute('%download_map('||&zoomlvl||','||loop_x||','||loop_y ||');');
    call execute('run;');
  end; end;
run;

/*Create an annotate data set to combine the tile images.*/
data anno_maps;
length function style $8;
xsys='2'; ysys='2'; when='b'; style='fit';
do loop_x=&min_x to &max_x;
  do loop_y=&min_y to &max_y;
imgpath="/cma/nshare/OSMMaps/daremap_&zoomlvl._"||trim(left(loop_x))||"_||trim
(left(loop_y))||".png";
  function='move';  x=loop_x;  y=loop_y+1; output;
  function='image'; x=loop_x+1; y=loop_y;  output;
  end; end;
run;


/*Since openstreetmaps starts at top/left flip the y-coordinate. */
data anno_maps;
set anno_maps end=last;
y=y*-1;
if last then do; output;
function='label';
  text="Map tiles by Thunderforest,Data by OpenStreetMap.";
xsys='1'; ysys='1';  x=100; y=0;
position='1'; when='a'; style='Arial/it';  size=.85;
end; output;
run;

PROC GANNO anno=anno_maps datasys;
run; quit;

```

## EXAMPLE SEVEN – TERRAIN MAP USING THUNDERFOREST TILE SET

In addition to the bicycle paths map tile data set, Thunderforest also makes available transport, landscape, outdoors, and a spinal map. The map below uses the landscape map tile data set for Dare County. Emphasis is on natural features. Although Dare County does not have any, in areas with high elevations, hills are labeled and contours are visible.




**Figure 7. Dare County Area Landscape with Thunderforest Map Tile Data.**

The only required change to the SAS code for this map from the map above is to the `_tileserv` macro variable as illustrated below:

```
filename osm_pic url
"http://a.tile.thunderforest.com/landscape/&zoom./&x./&y..png?apikey=3c2ebc25cf
ac43029c61cd8d0cad274b";
```


**EXAMPLE EIGHT – SPINAL MAP WITH THUNDERFOREST TILE SET**

The Spinal Thunderforest map tile set was inspired by the American mockumentary film, *This is Spinal Tap*, directed and co-written by Rob Reiner. The spinal map is the only map tile set illustrated here that supports a zoom level up to 11. It adds a tongue-in-cheek approach to mapping, while still demonstrating the power of custom map styles based on vector tiles. Common place names are re-interpreted as part of the overly dramatic theming. See below for a zoom of a small part of London:



**Figure 8. Illustration of Thunderforest Spinal Zoom from [www.thunderforest.com](http://www.thunderforest.com).**

The example below uses the Spinal map tile data set for the Dare County area. Note with the added zoom selected, less area is included in the map.



**Figure 9. Dare County Area with Thunderforest Spinal.**

Partial SAS Code below shows modifications from the above SAS code for the Bicycle Trails map.

```
/* Define a zoom level for the map tiles */
%let zoomlvl=11;
goptions xpixels=800 ypixels=600;

/* Download each map tile */
%macro download_map(zoom, x, y);
filename osm_pic url "http://a.tile.thunderforest.com/spinal-
map/&zoom./&x./&y..png?apikey=3c2ebc25cfac43029c61cd8d0cad274b";

/* min and max values for the tiles - increase maximum y */
data _null_;
  set test;
  call symput('min_x',int(x)-1);
  call symput('max_x',int(x)+1);
  call symput('min_y',int(y)-2);
  call symput('max_y',int(y)+1);
run;
```

## CONCLUSION

The availability of OpenStreetMap functionality in SAS through the OSM Annotate Generator provides tools to add streets and other landmarks to SAS/GRAPH PROC GMAP and other SAS mapping programs with only simple modifications. SAS provides the ability to combine these maps with choropleth and other SAS map features. Additional tile sets that work with OpenStreetMap are available for use that allow mapping of terrain, elevation and other features. Some tile sets can be used just to create fun maps. This paper illustrated the ability to create OSM maps with the features that you need for very diverse applications using the functionality provided by the SAS. This functionality includes the interface

to traditional OSM tile data sets as well as a growing number of free options from other developers such as Stamen, Thunderforest and CARTO. Using these map tile data sets is like selecting your favorite backslash pattern.

## REFERENCES

"About Dare County." Dare County. Archived from the original on 3 April 2012. <https://web.archive.org/web/20120403130938/http://www.darenc.com/general/mainabout.asp>. Accessed Sept 26, 2017.

Allen, Andy. Thunderforest. <https://www.thunderforest.com/>. Accessed June 14, 2018.

CARTO – Location Intelligence Software. <https://carto.com/>. Accessed June 20, 2018.

Massengill, D. 2013. "Google-Like' Maps in SAS®." SAS Global Forum 2013 Proceedings. Available at: <https://support.sas.com/resources/papers/proceedings13/377-2013.pdf>.

Okerson, B. 2018. "Mapping Roanoke Island Revisited: An OpenStreetMap (OSM) Solution." SAS Global Forum 2018 Proceedings. Available at: <https://www.sas.com/content/dam/SAS/support/en/sas-global-forum-proceedings/2018/2495-2018.pdf>.

"OpenStreetMap." <https://www.openstreetmap.org/about>. Accessed Sept 11, 2017.

"Stamen." <http://wiki.openstreetmap.org/wiki/Stamen>. Accessed Sept 26, 2017.

## ACKNOWLEDGMENTS

Special thanks go to Bob Allison and Marcia Suratt of SAS Institute for encouragement and assistance with some of the code in this paper.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Barbara B. Okerson, Ph.D., CPHQ, FAHM  
Independent Contractor  
bokerson1@charter.net

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.